Section 4. Program Memory

HIGHLIGHTS

This section of the manual contains the following topics:

4.1 PROGRAM MEMORY ADDRESS MAP

Figure 4-1 shows that the PIC24H devices have a 4M x 24-bit program memory address space. Three methods are available for accessing program space.

- Through the 23-bit (Program Counter) PC
- Through table read (TBLRD) and table write (TBLWT) instructions
- By mapping a 32-Kbyte segment of program memory into the data memory address space

The program memory map is divided into the user program space and the user configuration space. The user program space contains the Reset vector, interrupt vector tables, and program memory. The user configuration space contains nonvolatile configuration bits for setting device options and the device ID locations.

Figure 4-1: Example Program Memory Map



Note: The address boundaries for user Flash program memory will depend on the dsPIC33F device variant selected. For further details, refer to the appropriate device data sheet.

4.2 PROGRAM COUNTER

The Program Counter (PC) increments by 2 with the Least Significant bit (LSb) set to '0' to provide compatibility with data space addressing. Sequential instruction words are addressed in the 4M program memory space by PC<22:1>. Each instruction word is 24 bits wide.

The LSb of the program memory address (PC<0>) is reserved as a byte select bit for program memory accesses from data space that use Program Space Visibility or table instructions. For instruction fetches via the PC, the byte select bit is not required. Therefore, PC<0> is always set to '0'.

Figure 4-2 shows an instruction fetch example. Note that incrementing PC<22:1> by one is equivalent to adding 2 to PC<22:0>.

Figure 4-2: Instruction Fetch Example



Note 1: Increment of PC<22:1> is equivalent to PC<22:0>+2.

4.3 PROGRAM MEMORY ACCESS USING TABLE INSTRUCTIONS

The TBLRDL and TBLWTL instructions offer a direct method of reading or writing the least significant word (lsw) of any address within program space without going through data space, which is preferable for some applications. The TBLRDH and TBLWTH instructions are the only method whereby the upper 8 bits of a program word can be accessed as data.

4.3.1 Table Instruction Summary

A set of table instructions is provided to move byte- or word-sized data between program space and data space. The table read instructions are used to read from the program memory space into data memory space. The table write instructions allow data memory to be written to the program memory space.

Note:	Detailed code examples using table instructions are found in **Section 5. "Flash and EEPROM Programming"**.

The four available table instructions are:

- TBLRDL: Table Read Low
- TBLWTL: Table Write Low
- TBLRDH: Table Read High
- TBLWTH: Table Write High

For table instructions, program memory can be regarded as two 16-bit, word-wide address spaces residing side by side, each with the same address range as shown in Figure 4-3. This allows program space to be accessed as byte or aligned word addressable, 16-bit wide, 64-Kbyte pages (i.e., same as data space).

TBLRDL and TBLWTL access the least significant data word of the program memory, and TBLRDH and TBLWTH access the upper word. As program memory is only 24 bits wide, the upper byte from this latter space does not exist, although it is addressable. It is, therefore, termed the "phantom" byte.

Figure 4-3: High and Low Address Regions for Table Operations



4.3.2 Table Address Generation

Figure 4-4 shows how for all table instructions, a W register address value is concatenated with the 8-bit Data Table Page (TBLPAG) register, to form a 23-bit effective program space address plus a byte select bit. As there are 15 bits of program space address provided from the W register, the data table page size in program memory is, therefore, 32K words.

Figure 4-4: Address Generation for Table Operations



4.3.3 Program Memory Low Word Access

The TBLRDL and TBLWTL instructions are used to access the lower 16 bits of program memory data. The LSb of the W register address is ignored for word-wide table accesses. For byte-wide accesses, the LSb of the W register address determines which byte is read. Figure 4-5 demonstrates the program memory data regions accessed by the TBLRDL and TBLWTL instructions.

Figure 4-5: Program Data Table Access (lsw)



4.3.4 Program Memory High Word Access

The TBLRDH and TBLWTH instructions are used to access the upper 8 bits of the program memory data. Figure 4-6 shows how these instructions also support Word or Byte Access modes for orthogonality, but the high byte of the program memory data will always return '0'.

Figure 4-6: Program Data Table Access (MS Byte)



4.3.5 Data Storage in Program Memory

It is assumed that for most applications, the high byte (P<23:16>) is not used for data, making the program memory appear 16 bits wide for data storage. It is recommended that the upper byte of program data be programmed either as a NOP instruction, or as an illegal opcode value, to protect the device from accidental execution of stored data. The TBLRDH and TBLWTH instructions are primarily provided for array program/verification purposes and for applications that require compressed data storage.

4.3.6 Program Memory Access Using Table Instructions Example

The following example uses table instructions to erase the program memory page starting at the address 0x12000, and programs the values 0x123456 and 0x789ABC into addresses 0x12000 and 0x12002, respectively.

Note:	For more information on unlocking sequence, refer to **Section 5. "Flash Programming"**.

Example 4-1: Using Table Instructions to Access Program Memory

```
#define PM_ROW_ERASE 0x4042
#define PM_ROW_WRITE 0x4001
#define CONFIG_WORD_WRITE 0x4000

unsigned long Data;

/* Erase 512 instructions starting at address 0x12000 */
MemWriteLatch(0x1, 0x2000,0x0,0x0);
MemCommand(PM_ROW_ERASE);

/* Write 0x12345 into program address 0x12000 */
MemWriteLatch(0x1, 0x2000,0x0012,0x3456);
MemCommand(PM_ROW_WRITE);

/* Write 0x789ABC into program address 0x12002 */
MemWriteLatch(0x1, 0x2002,0x0078,0x9ABC);
MemCommand(PM_ROW_WRITE);

/* Read program addresses 0x12000 and 0x12002 */
Data = ReadLatch(0x1, 0x2000);
Data = ReadLatch(0x1, 0x2002);
```

```
;*********************************************************
;_MemWriteLatch:
;
;W0 = TBLPAG
;W1 = Wn
;W2 = WordHi
;W3 = WordLo
;no return values

_MemWriteLatch:
    mov   W0, TBLPAG
    tblwtl W3, [W1]
    tblwth W2, [W1]

    return
```

```
;*********************************************************
; _MemReadLatch:
;
;W0 = TBLPAG
;W1 = Wn
;return: data in W1:W0

_MemReadLatch:
    mov W0,TBLPAG
    tblrdl [W1],W0
    tblrdh [W1],W1

    return
```

```
;*********************************************************
; _MemCommand:
;
;W0 = NVMCON
;no return values

_WriteCommand:
    mov W0,NVMCON
    mov #0x55,W0;Unlock sequence
    mov W0,NVMKEY
    mov #0xAA,W0
    mov W0,NVMKEY
    bset  NVMCON,#WR
    nop            ;Required
    nop
Loop:btsc  NVMCON,#WR;Wait for write end
    bra   Loop

    return
```

4.4 PROGRAM SPACE VISIBILITY FROM DATA SPACE

The upper 32 Kbytes of the PIC24H data memory address space can optionally be mapped into any 16K word program space page. This mode of operation, called Program Space Visibility (PSV), provides transparent access of stored constant data from X data space without the need to use special instructions (i.e., TBLRD, TBLWT instructions).

4.4.1 PSV Configuration

Program Space Visibility is enabled by setting the PSV bit in the Core Control (CORCON<2>) register. A description of the CORCON register is found in **Section 2. "CPU"**.

When PSV is enabled, each data space address in the upper half of the data memory map will map directly into a program address (refer to 'Figure 4-7). The PSV window allows access to the lower 16 bits of the 24-bit program word. The upper 8 bits of the program memory data should be programmed to force an illegal instruction, or a NOP instruction, to maintain machine robustness. Table instructions provide the only method of reading the upper 8 bits of each program memory word.

Figure 4-8 shows how the PSV address is generated. The 15 LSbs of the PSV address are provided by the W register that contains the effective address. The Most Significant bit (MSb) of the W register is not used to form the address. Instead, the MSb specifies whether to perform a PSV access from program space or a normal access from data memory space. If a W register effective address of 0x8000 or greater is used, the data access will occur from program memory space when PSV is enabled. All data access occurs from data memory when the W register effective address is less than 0x8000.

Figure 4-8 shows how the remaining address bits are provided by the Program Space Visibility Page Address (PSVPAG<7:0>) register. The PSVPAG bits are concatenated with the 15 LSbs of the W register, holding the effective address to form a 23-bit program memory address. PSV can only be used to access values in program memory space. Table instructions must be used to access values in the user configuration space.

The LSb of the W register value is used as a byte select bit, which allows instructions using PSV to operate in Byte or Word mode.

Figure 4-7: Program Space Visibility Operation



Figure 4-8: Program Space Visibility Address Generation



4.4.2 PSV Timing

Instructions that use PSV requires two extra instruction cycles to complete execution, except for the following instructions that require only one extra cycle to complete execution:

- All MOV instructions including the MOV.D instruction

The additional instruction cycles are used to fetch the PSV data on the program memory bus.

4.4.2.1 USING PSV IN A REPEAT LOOP

Instructions that use PSV within a REPEAT loop eliminate the extra instruction cycle(s) required for the data access from program memory, therefore incurring no overhead in execution time. However, the following iterations of the REPEAT loop incur an overhead of two instruction cycles to complete execution:

- The first iteration
- The last iteration
- Instruction execution prior to exiting the loop due to an interrupt
- Instruction execution upon re-entering the loop after an interrupt is serviced

4.4.2.2 PSV AND INSTRUCTION STALLS

For more information about instruction stalls using PSV, refer to **Section 2. "CPU"**.

4.4.3 PSV Code Examples

Example 4-2: PSV Code Example in C

```
// PSV code example in C
// When defined as below the const string uses the PSV feature of dsPIC

const unsigned char hello[] = {"Hello World:\r\n"};
unsigned char *TXPtr;             // Transmit pointer

int main(void)
{
// Initialize the UART1
 U1MODE = 0x8000;
 U1STA = 0x0000;
 U1BRG = ((FCY/16)/BAUD) - 1;     // set baud rate = BAUD
 TXPtr = &hello[0];               // point to first char in string
 U1STAbits.UTXEN = 1;             // Initiate transmission
while (1)
    {
    while (*TXPtr)                // while valid char in string ...
        if (!U1STAbits.UTXBF)     // and buffer not full ...
            U1TXREG = *TXPtr++;   // transmit string via UART

    DelayNmSec(500);              // delay for 500 mS
    TXPtr = &hello[0];            // re-initialize pointer to first char
    }

}  // end main
```

Example 4-3: PSV Code Example in Assembly

```
.equ CORCONL, CORCON
    .section  .const, "r"
hello:
        .ascii "Hello World:\n\r\0"

      .global __reset  ;Declare the label for the start of code

    .text                     ;Start of Code section

__reset:

        clr     U1STA
        mov     #0x8000,W0     ; enable UART module
        mov     W0,U1MODE
        mov     #BR,W0         ; set baud rate using formula value
        mov     W0, U1BRG      ;   /
        bset    U1STA,#UTXEN   ; initiate transmission
Again:
        rcall   Delay500mSec   ; delay for 500 mS
        mov     #psvpage(hello),w0
        mov     w0, PSVPAG
        bset.b  CORCONL,#PSV
        mov     #psvoffset(hello),w0
TxSend:
        mov.b   [w0++], w1     ; get char in string
        cp      w1,#0          ; if Null
        bra     Z,Again        ; then re-initialize
BufferTest:
        btsc    U1STA,#UTXBF   ; see if buffer full
        bra     BufferTest     ; wait till empty
        mov     w1,U1TXREG     ; load value in TX buffer
        bra     TxSend         ; repeat for next char
```

Note: For additional information on MPLAB® C30 tools support for PSV, refer to the *"MPLAB® C30 Managed PSV Pointers"* document in the C30 installation directory.

4.5 PROGRAM MEMORY WRITES

The PIC24H family of devices contains internal program Flash memory for executing user code. There are two methods by which the user application can program this memory:

- Run-Time Self Programming (RTSP)
- In-Circuit Serial Programming™ (ICSP™)

RTSP is accomplished using TBLWT instructions. ICSP is accomplished using the SPI interface and integral bootloader software. For further details about RTSP, refer to **Section 5. "Flash and EEPROM Programming"**. ICSP specifications can be downloaded from the Microchip Technology web site (www.microchip.com).

4.6 CONTROL REGISTER

The register described in this section controls PSV.

Register 4-1: CORCON: Core Control Register

U-0	U-0	U-0	U-0	U-0	U-0	U-0	U-0
—	—	—	—	—	—	—	—
bit 15							bit 8

U-0	U-0	U-0	U-0	R/C-0	R/W-0	U-0	U-0
—	—	—	—	IPL3	**PSV**	—	—
bit 7							bit 0

Legend:			
R = Readable bit	W = Writable bit	U = Unimplemented bit, read as ‘0’	
-n = Value at POR	‘1’ = Bit is set	‘0’ = Bit is cleared	x = Bit is unknown

bit 15-3	**Not used by the Program Memory** (For full descriptions of the CORCON bits, refer to **Section 2. “CPU”**)
bit 2	**PSV:** Program Space Visibility in Data Space Enable bit 1 = Program space visible in data space 0 = Program space not visible in data space
bit 1-0	**Not used by the Program Memory** (For full descriptions of the CORCON bits, refer to **Section 2. “CPU”**)

4.7 RELATED APPLICATION NOTES

This section lists application notes that are related to this section of the manual. These application notes may not be written specifically for the PIC24H device family, but the concepts are pertinent and could be used with modification and possible limitations. The current application notes related to the Program Memory module are:

Title	Application Note #
No related application notes at this time.	

Note: For additional Application Notes and code examples for the PIC24H device family, visit the Microchip web site (www.microchip.com).

4.8 REVISION HISTORY

Revision A (April 2007)

This is the initial released version of this document.